Associação de tubos em série e em paralelo. Uso do Excel.

Engº Luiz Camargo

O presente texto tem por objetivo a análise de alguns aspectos da associação de condutos em série e em paralelo a serem considerados no dimensionamento de tubulações que envolvam tais associações, que aqui serão examinados com a utilização das equações de Darcy-Weisbach, Colebrook-White e Swamee-Jain, fazendo uso do Excel, inclusive da conhecida ferramenta "Solver", o onde são previamente conhecidos o comprimento, diâmetro, rugosidade dos tubos e a viscosidade cinemática do líquido em escoamento. Então, são duas as situações na análise aqui proposta: a determinação da vazão e a determinação da carga piezométrica, sendo que a determinação de qualquer uma dessas variáveis a outra deverá ser conhecida. Problemas típicos serão resolvidos.

1 - Condutos em Paralelo

Diz-se que dois ou mais condutos estão em paralelo quando suas extremidades a montante estão reunidas num mesmo ponto e as extremidades a jusante reunidas em outro ponto, de modo a permitir que a vazão se divida entre os condutos e posteriormente torne a se reunificar.



A característica importante de um sistema de condutos em paralelo é que as perdas de carga em cada um deles é a mesma, enquanto que as vazões são somadas.

Na associação de condutos em paralelo, dois tipos de problema serão analisados: a determinação da vazão Q para uma altura H dada, e o cálculo da perda de carga para uma vazão conhecida (Lencastre, 1983). Considera-se H a diferença de cargas piezométricas entre os pontos A e B da figura acima.

1.1 - Determinação da altura quando conhecida a vazão.

As perdas de carga em tubos são calculadas pela equação de Darcy-Weisbach:

$$h_f = \frac{8 fLQ^2}{g\pi^2 D^5} \quad (1)$$

Considerando a perda de carga a mesma em cada um dos condutos, isto é, sendo a perda de carga comum aos n condutos entre os pontos A e B, então H é calculado com:

$$H = \frac{8 f_1 L_1 Q_1^2}{g\pi^2 D_1^5} = \frac{8 f_2 L_2 Q_2^2}{g\pi^2 D_2^5} = \ldots = \frac{8 f_n L_n Q_n^2}{g\pi^2 D_n^5} \quad (2)$$

Sendo que a vazão total Q, como já dito anteriormente, é dada pela soma das vazões dos tubos em paralelo:

$$Q = Q_1 + Q_2 + Q_3 + \ldots + Q_n \quad (3)$$

onde:

H = carga piezométrica entre os pontos A e B (m).
Q = vazão total (m^3/s).
Q_i = vazão em cada conduto (m^3/s).

f_i = fator de atrito de Darcy-Weisbach em cada conduto (adimensional).
L_i = comprimento de cada conduto (m).
D_i = diâmetro de cada conduto (m).
g = aceleração da gravidade local (m/s^2).
i = 1, 2, 3, ... $\boldsymbol{n}$.
$\boldsymbol{n}$ = número de condutos em paralelo.

Se a carga piezométrica H é a mesma para todos os condutos, então a diferença entre as cargas de dois quaisquer deles é zero. Portanto, das Eqs. (2) pode-se dizer que:

$$\frac{8f_1L_1Q_1^{\,2}}{g\pi^2D_1^{\,5}}-\frac{8f_2L_2Q_2^{\,2}}{g\pi^2D_2^{\,5}}=0 \tag{4}$$

$$\frac{8f_2L_2Q_2^{\,2}}{g\pi^2D_2^{\,5}}-\frac{8f_3L_3Q_3^{\,2}}{g\pi^2D_3^{\,5}}=0 \tag{5}$$

... ...

$$\frac{8f_{n-1}L_{n-1}Q_{n-1}^{\;\;2}}{g\pi^2D_{n-1}^{\;\;5}}-\frac{8f_nL_nQ_n^{\,2}}{g\pi^2D_n^{\,5}}=0 \tag{6}$$

Para cálculo do fator de atrito, Swamee e Jain (1976) desenvolveram a seguinte expressão:

$$f=0,25\left[\log_{10}\left(\frac{k}{3,7D}+\frac{5,74}{R_e^{\,0,9}}\right)\right]^{-2} \tag{7}$$

O Número de Reynolds, por definição, é dado por:

$$R_e=\frac{4Q}{\pi D\nu} \tag{8}$$

Substituindo a Eq. (8) na Eq. (7) e rearrumando, em cada conduto tem-se:

$$f_i=0,25\left[\log_{10}\left(\frac{k_i}{3,7D_i}+4,618\left(\frac{D_i\nu}{Q_i}\right)^{0,9}\right)\right]^{-2} \tag{9}$$

com:

k_i = rugosidade equivalente da parede de cada conduto (m)
ν = viscosidade cinemática do líquido em escoamento (m^2/s)

Então, substituindo a Eq. (9) nas Eqs. (4) a (6) e realizando operações, tem-se:

$$Q_1\sqrt{L_1D_2^{\,5}}\,\log_{10}\left[\frac{k_2}{3,7D_2}+4,618\left(\frac{D_2\nu}{Q_2}\right)^{0,9}\right]-Q_2\sqrt{L_2D_1^{\,5}}\,\log_{10}\left[\frac{k_1}{3,7D_1}+4,618\left(\frac{D_1\nu}{Q_1}\right)^{0,9}\right]=0 \tag{10}$$

$$Q_2\sqrt{L_2D_3^{\,5}}\,\log_{10}\left[\frac{k_3}{3,7D_3}+4,618\left(\frac{D_3\nu}{Q_3}\right)^{0,9}\right]-Q_3\sqrt{L_3D_2^{\,5}}\,\log_{10}\left[\frac{k_2}{3,7D_2}+4,618\left(\frac{D_2\nu}{Q_2}\right)^{0,9}\right]=0 \tag{11}$$

. . .

$$Q_{n-1}\sqrt{L_{n-1}D_n^{\,5}}\,\log_{10}\left[\frac{k_n}{3,7D_n}+4,618\left(\frac{D_n\nu}{Q_n}\right)^{0,9}\right]-Q_n\sqrt{L_nD_{n-1}^{\;\;5}}\,\log_{10}\left[\frac{k_{n-1}}{3,7D_{n-1}}+4,618\left(\frac{D_{n-1}\nu}{Q_{n-1}}\right)^{0,9}\right]=0 \tag{12}$$

Se a vazão total Q é a soma das vazões dos $\boldsymbol{n}$ condutos, então:

$$Q - Q_1 - Q_2 - Q_3 - \ldots - Q_n = 0 \tag{13}$$

Então, tem-se aqui um sistema de $\boldsymbol{n}$ equações, Eqs. (10), (11), (12) e (13), com $\boldsymbol{n}$ incógnitas, Q_1, Q_2, Q_3, **...** Q_n, mas, como se vê, trata-se de equações não lineares. De um modo geral a solução de um sistema desta natureza envolve algum processo iterativo de aproximações sucessivas. O mais apropriado é utilizar o cálculo automatizado, por processo computacional, tal como faz a conhecida ferramenta "Solver" do Excel, que será utilizada no presente caso. A solução com esta ferramenta consiste basicamente em utilizar o primeiro membro da Eq. (10) como célula objetivo na caixa "Definir Objetivo", marcando a opção "Valor de" e adotando para esta opção o valor zero. Na caixa "Alterando Células Variáveis" insira a referência para o intervalo das células variáveis destinadas às incógnitas. Na caixa "Sujeito às Restrições" insira as referências relativas às células dos primeiros membros das demais equações, isto é, Eqs. (11), (12) e (13), restringindo-os a que também sejam zero. A solução do problema, portanto, serão os valores de Q que, simultaneamente, tornam nulos os primeiros membros de **todas** as equações do sistema aqui definido. Ao se clicar no botão "Resolver", automaticamente o Solver encontrará esses valores, como mostra a solução do problema apresentado a seguir. Como nas equações do sistema as incógnitas aparecem também no denominador, é conveniente atribuir às mesmas um valor inicial maior que zero, digamos 0,1, com vistas a evitar o erro de divisão por zero, que travaria a resolução.

A diferença entre as cargas piezométricas dos pontos extremos, de reunião dos condutos, pode ser obtida substituindo a Eq. (9) na Eq. (2):

$$H = \frac{2L_1 Q_1^{\,2}}{g\pi^2 D_1^{\,5}} \left[\log_{10}\left(\frac{k_1}{3,7D_1} + 4,618\left(\frac{D_1 \nu}{Q_1} \right)^{0,9} \right) \right]^{-2} \tag{14}$$

PROBLEMA N° 1 (exemplo 11.6, Streeter e Wylie, 1985, pág. 441):

Na figura ao lado, L_1 = 914,4 m, D_1 = 0,3048 m, k_1 = 0,0003048 m; L_2 = 609,6 m, D_2 = 0,2032 m, k_2 = 0,00003048 m; L_3 = 1219,2 m, D_3 = 0,4064 m, k_3 = 0,0002438 m; ν = 0,000002787 m²/s. Para uma vazão total de 0,34 m³/s, determine a vazão que escoa em cada tubo (Obs: original em unid. inglesas).





	A	B	C	D	E
1	**Tubos em paralelo - Dado Q encontrar Q_1, Q_2, ... , Q_n e H - Uso da função Solver:**				
2	D_1 (m) =	0,30480		$n = 3$	
3	D_2 (m) =	0,20320			
4	D_3 (m) =	0,40640			
5	L_1 (m) =	914,40		Equação 10 =	0,0000
6	L_2 (m) =	609,60		Equação 11 =	0,0000
7	L_3 (m) =	1219,20		Equação 13 =	0,0000
8	k_1 (m) =	0,0003048		Q_1 (m³/s) =	0,1012
9	k_2 (m) =	0,0000305		Q_2 (m³/s) =	0,0487
10	k_3 (m) =	0,0002438		Q_3 (m³/s) =	0,1901
11	ν (m²/s) =	0,000002787		H (m) (Eq. 14) =	6,353
12	g (m/s²) =	9,806			
13	Q (m³/s) =	0,340			

Resposta do livro:

Q_1 = 0,10138 m³/s
Q_2 = 0,04871 m³/s
Q_3 = 0,18974 m³/s
H = 6,3398 m

1.2 - Determinação da vazão quando conhecida a altura.

Este caso equivale, na realidade, à simples determinação da vazão em condutos individuais, pois a perda de carga conhecida possibilita a determinação imediata das vazões em cada um dos tubos, cuja soma resulta na vazão total.

A Eq. (1) pode ser escrita na conveniente forma:

$$\frac{1}{\sqrt{f}} = \sqrt{\frac{8Q^2}{\pi^2 g D^5 H/L}} \tag{15}$$

Aqui, para cálculo do fator de atrito, a equação de Colebrook-White (1938/1939) é mais adequada:

$$\frac{1}{\sqrt{f}} = -2\log_{10}\left(\frac{k}{3{,}7D} + \frac{2{,}51}{R_e\sqrt{f}}\right) \tag{16}$$

Considerando as perdas de carga singulares como comprimentos equivalentes já adicionados aos comprimentos dos condutos e eliminando f pela substituindo as Eqs. (8) e (15) na Eq. (16), a vazão em cada conduto é dada com a seguinte expressão:

$$Q_i = -\frac{\pi}{2}\sqrt{2gD_i^5\, H/L_i}\cdot\log_{10}\left(\frac{k_i}{3{,}7D_i} + \frac{2{,}51\nu}{\sqrt{2gD_i^3\, H/L_i}}\right) \tag{17}$$

na qual as variáveis envolvidas já foram definidas anteriormente.

A Eq. (17) levada ao Excel, com os dados do problema, possibilita a solução automática, como mostra a planilha adiante apresentada.

PROBLEMA N° 2 (problema 11.20, Evett e Liu, 1989, pág. 287):

Três condutos estão em paralelo com uma perda de carga total de 20,3 m. Os dados dos condutos são: L_1 = 100 m, D_1 = 0,08 m, k_1 = 0.00024 m; L_2 = 150 m, D_2 = 0,06 m, k_2 = 0.00012 m; L_3 = 80 m, D_3 = 0,04 m, k_3 = 0.00020 m. O fluido é água, ρ = 1000 kg /m^3 e ν = 1,02×10^{-6} m^2/s. Calcule a vazão total, desprezando as perdas localizadas.

A19

	A	B	C	D	E	F	G	H
1	**Tubos em série. Dado H, encontrar Q :**							
2	D_1 (m) =	0,080		**n = 3**				
3	D_2 (m) =	0,060						
4	D_3 (m) =	0,040		Q_1 (m^3/s) =	0,01737			
5	L_1 (m) =	100,00		Q_2 (m^3/s) =	0,00720			
6	L_2 (m) =	150,00		Q_3 (m^3/s) =	0,00317			
7	L_3 (m) =	80,00		Q_{tot} (m^3/s) =	0,02774			
8	k_1 (m) =	0,00024						
9	k_2 (m) =	0,00012						
10	k_3 (m) =	0,00020						
11	ν (m^2/s) =	0,00000102						
12	g (m/s^2) =	9,807						
13	H (m^3/s) =	20,30						
14								

Resposta do livro: Q_{tot} = 0,01736 + 0,007193 + 0,003169 = 0,02772 m^3/s

2 - Condutos em Série

Diz-se que dois ou mais condutos de comprimentos, diâmetros e rugosidades distintas estão em série quando os mesmos estão interligados por suas extremidades e permitem o escoamento contínuo do fluido através de cada um deles, um após o outro.

A característica importante de um sistema de condutos em série é que a vazão Q é a mesma em todos os condutos, enquanto que as perdas de carga são somadas.



Tipicamente a análise hidráulica de tubos em série se depara com dois tipos de problema: determinar a altura H para uma vazão Q dada, e encontrar a vazão para uma altura conhecida (Lencastre, 1983).

2.1 - Determinação da altura quando conhecida a vazão.

Considerando as perdas de carga singulares como comprimentos equivalentes já adicionados aos comprimentos dos condutos e aplicando a Eq. (1) de Darcy-Weisbach a cada um dos n condutos em série, H é obtido com:

$$H = \frac{8Q^2}{g\pi^2}\left(\frac{f_1 L_1}{D_1^5} + \frac{f_2 L_2}{D_2^5} + \cdots + \frac{f_n L_n}{D_n^5}\right) \tag{18}$$

na qual as variáveis envolvidas já foram definidas anteriormente.

Portanto a altura H é calculada com a Eq. (18) a qual receberá os valores de f obtidos com a Eq. (9).

PROBLEMA N° 3 (problema n° 10-24, Evett e Liu, 1989, pág. 275):

Dois reservatórios são conectados por três condutos de ferro fundido não revestido (k = 0,00026 m) em série, sendo que: L_1 = 300 m, D_1 = 0,20 m; L_2 = 400 m, D_2 = 0,30 m; L_3 = 1200 m, D_3 = 0,45 m. Se a vazão for de 360 m^3/h (0,10 m^3/s) de água a 20 °C (ν = 1,02×10^{-6} m^2/s), determine a diferença de altitude dos reservatórios.

B29 | fx

	A	B	C	D	E	F	G	H	I	J
1	Tubos em série. Dado Q, encontrar H:									
2	D_1 (m) =	0,200		n = 3						
3	D_2 (m) =	0,300								
4	D_3 (m) =	0,450		f_1 =	0,02146					
5	L_1 (m) =	300,00		f_2 =	0,01988					
6	L_2 (m) =	400,00		f_3 =	0,01886					
7	L_3 (m) =	1200,00		H (m) =	20,35					
8	k_1 (m) =	0,00026								
9	k_2 (m) =	0,00026								
10	k_3 (m) =	0,00026								
11	ν (m^2/s) =	0,00000102								
12	g (m/s^2) =	9,807								
13	Q (m^3/s) =	0,1000								

Resposta do livro: H = 20,37 m.

2.2 - Determinação da vazão quando conhecida a altura.

Considerando as perdas de carga singulares como comprimentos equivalentes já adicionados aos comprimentos dos condutos, aqui a vazão pode ser calculada explicitando Q na Eq. (18), obtendo-se:

$$Q = \frac{\pi}{4}\sqrt{2gH} \cdot \left(\frac{f_1 L_1}{D_1^{\,5}} + \frac{f_2 L_2}{D_2^{\,5}} + \cdots + \frac{f_n L_n}{D_n^{\,5}} \right)^{-1/2} \qquad (19)$$

Substituindo as Eqs. (8) e (15) na Eq. (16), e realizando operações, tem-se:

$$f_i = \left[-2\log_{10}\left(\frac{k_i}{3,7D_i} + \frac{2,51\nu}{\sqrt{2gD_i^{\,3}\, H/L_i}} \right) \right]^{-2} \qquad (20)$$

onde todas as variáveis envolvidas já foram definidas anteriormente.

Portanto a vazão Q é calculada com a Eq. (19) a qual receberá os valores de f obtidos com a Eq. (20).

PROBLEMA N° 4 (problema n° 10-14, Evett e Liu, 1989, pág. 271):

Considere na figura ao lado que os condutos 1, 2 e 3 são de ferro fundido novo (k = 0,00026 m) em série, tendo comprimento e diâmetro interno, respectivamente, de 300 m e 30 cm, 150 m e 20 cm, e 250 m e 25 cm, conduzindo água a 15 °C (ν = 1,16 x 10^{-6} m^2/s). Se H = 10 m, despreze as singularidades e calcule a vazão.



E7 =PI()/4*RAIZ(2*B12*B13)*(E4*B5/B2^5+E5*B6/B3^5+E6*B7/B4^5)^-0,5

	A	B	C	D	E	F
1	Tubos em série. Dado H, encontrar Q :					
2	D_1 (m) =	0,300		n = 3		
3	D_2 (m) =	0,200				
4	D_3 (m) =	0,250		f_1 =	0,01938	
5	L_1 (m) =	300,00		f_2 =	0,02137	
6	L_2 (m) =	150,00		f_3 =	0,02027	
7	L_3 (m) =	250,00		Q (m) =	0,0829	
8	k_1 (m) =	0,00026				
9	k_2 (m) =	0,00026				
10	k_3 (m) =	0,00026				
11	ν (m^2/s) =	0,00000116				
12	g (m/s^2) =	9,807				
13	H (m^3/s) =	10,00				
14						

Resposta do livro: Q = 0,0830 m^3/s.

Bibliografia:

1 - Colebrook, C.F. "Turbulent Flow in Pipes, with Particular Reference to the Transition Region Between the Smooth and Rough Pipes", Journal of the Institution of Civil Engineers, v. 11, p. 133-156, 1938/1939.
2 - Evett, J.B. & Liu, C. "2500 Solved Problems in Fluid Mechanics & Hydraulics", McGraw-Hill Schaum's, NY, 1989.
3 - Lencastre, A. "Hidráulica Geral", Hidroprojecto, Lisboa, 1983.
4 - Streeter, V.L. & Wylie, E.B. "Fluid Mechanics", McGraw-Hill, 8th Edition, NY, 1985.
5 - Swamee, P.K. & Jain, A.K. "Explicit Equations for Pipe-Flow Problems", Journal of the Hyd. Division, p. 657-664, May, 1976.

LC, Vitória, Dez/2023.

Apêndice:

Abaixo as equações utilizadas no texto, convertidas para a notação do Excel.

Problema N° 1:

Eq. (10)
=E8*RAIZ(B5*B3^5)*LOG10(B9/(3,7*B3)+4,618*(B3*B11/E9)^0,9)-E9*RAIZ(B6*B2^5)*LOG10(B8/(3,7*B2)+4,618*(B2*B11/E8)^0,9)

Eq. (11)
=E9*RAIZ(B6*B4^5)*LOG10(B10/(3,7*B4)+4,618*(B4*B11/E10)^0,9)-E10*RAIZ(B7*B3^5)*LOG10(B9/(3,7*B3)+4,618*(B3*B11/E9)^0,9)

Eq. (13)
=B13-E8-E9-E10

Eq. (14)
=2*B5*E8^2/(B12*PI()^2*B2^5)*(LOG10(B8/(3,7*B2)+4,618*(B2*B11/E8)^0,9))^-2

Problema N° 2:

Eq. (17)(1)
=-PI()/2*RAIZ(2*B12*B2^5*B13/B5)*LOG10(B8/(3,7*B2)+2,51*B11/RAIZ(2*B12*B2^3*B13/B5))

Eq. (17)(2)
=-PI()/2*RAIZ(2*B12*B3^5*B13/B6)*LOG10(B9/(3,7*B3)+2,51*B11/RAIZ(2*B12*B3^3*B13/B6))

Eq. (17)(3)
=-PI()/2*RAIZ(2*B12*B4^5*B13/B7)*LOG10(B10/(3,7*B4)+2,51*B11/RAIZ(2*B12*B4^3*B13/B7))

Problema N° 3:

f1
=0,25*(LOG10(B8/(3,7*B2)+4,618*(B2*B11/B13)^0,9))^-2

f2
=0,25*(LOG10(B9/(3,7*B3)+4,618*(B3*B11/B13)^0,9))^-2

f3
=0,25*(LOG10(B10/(3,7*B4)+4,618*(B4*B11/B13)^0,9))^-2

H
=8*B13^2/(B12*PI()^2)*(E4*B5/B2^5+E5*B6/B3^5+E6*B7/B4^5)

Problema N° 4:

f1
=(-2*LOG10(B8/(3,7*B2)+2,51*B11/RAIZ(2*B12*B2^3*B13/B5)))^-2

f2
=(-2*LOG10(B9/(3,7*B3)+2,51*B11/RAIZ(2*B12*B3^3*B13/B6)))^-2

f3
=(-2*LOG10(B10/(3,7*B4)+2,51*B11/RAIZ(2*B12*B4^3*B13/B7)))^-2

Q
=PI()/4*RAIZ(2*B12*B13)*(E4*B5/B2^5+E5*B6/B3^5+E6*B7/B4^5)^-0,5